Computer Science 355
Modern Computer Algebra

Assignment 3

solutions

Problem 1

Prove the following identity for $n \geq 2$

$$\sum_{k=0}^{n} \left(18 k^2 - 9 k n + 3 k - 8 n - 12\right) \binom{n+4}{3k-n} = 2 (-1)^n (n+3)(n+4)$$

Demonstrate each step of the algorithm and provide a certificate.

■

Divide the identity by the right-hand side

```
F[n_, k_] := (18 k^2 - 9 k n + 3 k - 8 n - 12)
   Binomial[n + 4, 3 k - n] / (2 (-1)^n (n + 3) (n + 4))
```

```
s[k_] := F[n + 1, k] - F[n, k]
```

```
s[k + 1]
--------  // FunctionExpand // Factor
  s[k]
```

$$-\left((-6+3k-2n)(-5+3k-2n)(-4+3k-2n)\left(-6+99k+12k^2+81k^3+54k^4-55n+75kn-99k^2n-81k^3n-42n^2+63kn^2+45k^2n^2-17n^3-9kn^3\right)\right) \Big/ \left((1+3k-n)(2+3k-n)(3+3k-n)\left(-120+102k+93k^2-135k^3+54k^4-148n+30kn+144k^2n-81k^3n-60n^2-27kn^2+45k^2n^2-8n^3-9kn^3\right)\right)$$

```
p[k_] := (-120 + 102 k + 93 k^2 - 135 k^3 + 54 k^4 - 148 n + 30 k n +
    144 k^2 n - 81 k^3 n - 60 n^2 - 27 k n^2 + 45 k^2 n^2 - 8 n^3 - 9 k n^3)
q[k + 1 ] = - (-6 + 3 k - 2 n) (-5 + 3 k - 2 n) (-4 + 3 k - 2 n) ;
r[k_ ] := (-2 + 3 k - n) (-1 + 3 k - n) (3 k - n)
```

```
q[k + 1] f[k] - r[k] f[k - 1] - p[k]
```

$$120-102k-93k^2+135k^3-54k^4+148n-30kn-144k^2n+81k^3n+60n^2+27kn^2-45k^2n^2+8n^3+9kn^3-(-2+3k-n)(-1+3k-n)(3k-n)\,f[-1+k]+(-5+3k-2n)(-4+3k-2n)(6-3k+2n)\,f[k]$$

```
Collect[% /. f[k_] :> a0 + a1 k, k, Factor];
CoefficientList[%, k]
```

$\{(2+n)\left(60+60\,a0+44\,n+45\,a0\,n-a1\,n+8\,n^2+9\,a0\,n^2-a1\,n^2\right),$
$-3\left(34+76\,a0-42\,a1+10\,n+66\,a0\,n-56\,a1\,n-9\,n^2+15\,a0\,n^2-24\,a1\,n^2-3\,n^3-3\,a1\,n^3\right),$
$3\left(-31+54\,a0-85\,a1-48\,n+27\,a0\,n-75\,a1\,n-15\,n^2-15\,a1\,n^2\right),$
$-27\,(-5+2\,a0-7\,a1-3\,n-3\,a1\,n),\ -54\,(1+a1)\}$

```
Solve[% == 0, {a0, a1}]
```

{{a0 → -1, a1 → -1}}

```
f[k_] := -1 - k
```

$G[n\_,\ k\_] := \frac{r[k]}{p[k]}\, f[k-1]\ s[k]$

The algorithm returns the following certificate

$\frac{G[n,\ k]}{F[n,\ k]}$ `// FunctionExpand // Factor`

$(3\,k\,(-2+3\,k-n)\,(-1+3\,k-n)\,(3\,k-n))\,\big/\,\left((-6+3\,k-2\,n)\,(-5+3\,k-2\,n)\left(-12+3\,k+18\,k^2-8\,n-9\,k\,n\right)\right)$

It remains to check the identity for a single value of $n = 2$.

$n = 2;\ \sum_{k=0}^{n}\left(18\,k^2-9\,k\,n+3\,k-8\,n-12\right)\,\text{Binomial}[n+4,\ 3\,k-n]$

60

$2\,(-1)^n\,(n+3)\,(n+4)$

60

Problem 2

Find a recurrence relation for the following sum

$$\sum_k \left(k^2 - 9\,k + 4\right)\binom{n}{k}$$

Demonstrate each step of Zeilberger's algorithm.

■

```
F[n_, k_] := Binomial[n, k] (k^2 - 9 k + 4)
s[k_] := d1 F[n + 1, k] + d0 F[n, k]
s[k + 1] / s[k] // FunctionExpand // Simplify

-(((-4 - 7 k + k^2) (-1 + k - n) (d0 (k - n) - d1 (1 + n))) /
   ((1 + k) (4 - 9 k + k^2) (d0 (-1 + k - n) - d1 (1 + n))))

p[k_] := (d0 (-1 + k - n) - d1 (1 + n)) (4 - 9 k + k^2)
q[k_] := - (-1 + (k - 1) - n)
r[k_] := k

Table[PolynomialGCD[q[j], r[j + t]], {t, 0, 15}]

{1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1}

p[k + 1] q[k + 1] / (p[k] r[k + 1]) - s[k + 1] / s[k] // FullSimplify

0

Exponent[q[k + 1] + r[k], k] > Exponent[q[k + 1] - r[k], k]

False

Exponent[p[k], k] - Exponent[q[k + 1] - r[k], k]

2
```

```
Clear[a, b, c];
q[k+1] f[k] - r[k] f[k-1] - p[k] /. f[k_] → a + b k + c k^2;
Collect[%, k];
CoefficientList[%, k];
Map[Equal[#, 0] &, %]
```

```
{a + 4 d0 + a n + 4 d0 n + 4 d1 (1 + n) == 0,
 -2 a + 2 b - c - 13 d0 + b n - 9 d0 n - 9 d1 (1 + n) == 0,
 -2 b + 3 c + 10 d0 + c n + d0 n + d1 (1 + n) == 0, -2 c - d0 == 0}
```

```
sols = Solve[%, {d0, d1, a, b, c}] // Simplify
```

— Solve::svars : Equations may not give solutions for all "solve" variables. >>

$$\left\{\left\{d1 \to -\frac{d0\left(16-17n+n^2\right)}{2(-15+n)n},\ a \to -\frac{2\,d0\left(-16-13n+n^2\right)}{(-15+n)n},\ b \to \frac{d0\left(-8-127n+9n^2\right)}{2(-15+n)n},\ c \to -\frac{d0}{2}\right\}\right\}$$

```
Clear[F];
d0 F[n, k] + d1 F[n+1, k] /. sols[[1]]
```

```
d0 F[n, k] - (d0 (16 - 17 n + n^2) F[1 + n, k]) / (2 (-15 + n) n)
```

The algorithm returns the following equation

$$\left(n^2 - 17\,n + 16\right) S(n+1) - 2\,(n-15)\,n\,S(n) = 0$$

Problem 3

Let $n \geq 0$ be any integer and let k be any integer such that $k \geq n+1$. Find a recurrence for

$$\sum_{j=0}^{n} \frac{(-1)^j \binom{k}{j} \binom{-j+k-1}{n-j}}{j+1}$$

Demonstrate each step of Zeilberger's algorithm.

■

First note that

$$\sum_{j=0}^{n} \neq \sum_{j=-\infty}^{\infty}$$

Indeed, let $k = n+1$ and $j = n+1$

```
k = n + 1; j = n + 1; (-1) ^ j Binomial[k, j] Binomial[k - 1 - j, n - j] / (j + 1)
```

$$\frac{(-1)^{1+n}}{2+n}$$

Therefore, Zeilberger's algorithm will return a non-homogeneus difference equation.

```
Clear[k, n, j];
F[n_, j_] := (-1) ^ j Binomial[k, j] Binomial[k - 1 - j, n - j] / (j + 1)

s[j_] := Sum[d[p] F[n + p, j], {p, 0, 1}]

s[j + 1] / s[j] // FunctionExpand // Factor
```

```
((j - k) (-1 + j - n) (j d[0] - n d[0] + d[1] - k d[1] + n d[1])) /
  ((2 + j) (1 + j - k) (-d[0] + j d[0] - n d[0] + d[1] - k d[1] + n d[1]))
```

```
p[j_] := -d[0] + j d[0] - n d[0] + d[1] - k d[1] + n d[1]
q[j_] := (j - 1 - k) (-2 + j - n) ;
r[j_] := (1 + j) (j - k)

Table[PolynomialGCD[q[j], r[j + t]], {t, 0, 10}]
```

```
{1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1}
```

```
p[j+1] q[j+1]     s[j+1]
-------------- - ------- // FullSimplify
 p[j] r[j+1]       s[j]
```

```
0
```

```
q[j+1] f[j] - r[j] f[j-1] - p[j]
```

```
d[0] - j d[0] + n d[0] - d[1] + k d[1] - n d[1] -
  (1 + j) (j - k) f[-1 + j] + (j - k) (-1 + j - n) f[j]
```

```
Collect[% /. f[j_] → c, j, Factor];
# == 0 & /@ CoefficientList[%, j]
```

```
{2 c k + c k n + d[0] + n d[0] - d[1] + k d[1] - n d[1] == 0, -2 c - c n - d[0] == 0}
```

```
Solve[%, {c, d[0], d[1]}]
```

— *Solve::svars : Equations may not give solutions for all "solve" variables.* ≫

```
{{d[0] → -c (2 + n), d[1] → -c (2 + n)}}
```

```
Clear[F]; s[j] /. %[[1]]
```

```
-c (2 + n) F[n, j] - c (2 + n) F[1 + n, j]
```

This leads to the following recurrence

$$\sum_{j=0}^{n} (F[n, j] + F[n+1, j]) == 0$$

We sum it up as in Celine's algorithms

$$\sum_{j=0}^{n} F[n, j] + \sum_{j=0}^{n} F[n+1, j] == S[n] - F[n+1, n+1] + \sum_{j=0}^{n+1} F[n+1, j] ==$$

$$S[n] + S[n+1] - F[n+1, n+1]$$

Computing $F[n+1, n+1]$, we obtain

$$S[n] + S[n+1] == \frac{(-1)^{n+1} \text{Binomial}[k, 1+n]}{2+n}$$